Maple-assisted proof of formula for A208554

Robert Israel

27 Sep 2018

There are $2^{14} = 16384$ possible configurations for a 2×7 sub-array, but not all can arise as we need to avoid 0 0 0 and 0 0 1 horizontally.

We encode these configurations as lists in the order $\begin{bmatrix} x_1 & x_2 & x_3 & x_4 & x_5 & x_6 & x_7 \\ x_8 & x_9 & x_{10} & x_{11} & x_{12} & x_{13} & x_{14} \end{bmatrix}$. The possible rows are as follows.

```
> Rowconfigs:= select(A -> andmap(t -> A[t..t+2] <> [0,0,0] and A
  [t..t+2] <> [0,0,1], [$1..5]), [seq(convert(x,base,2)[1..7],x=
  2^7..2^8-1)]);
```

Rowconfigs := [[1, 0, 1, 0, 1, 0, 0], [0, 1, 1, 0, 1, 0, 0], [1, 1, 1, 0, 1, 0, 0], [0, 1, 0, 1, 1, 0, 0], [1, 1, 0, 1, 1, 0, 0], [1, 0, 1, 1, 1, 0, 0], [0, 1, 1, 1, 1, 0, 0], [1, 1, 1, 1, 1, 0, 0], [0, 1, 0, 1, 0, 1, 0], [1, 1, 0, 1, 0, 1, 0], [1, 0, 1, 1, 0, 1, 0], [0, 1, 1, 1, 0, 1, 0], [1, 1, 1, 1, 0, 1, 0], [1, 0, 1, 0, 1, 1, 0], [0, 1, 1, 0, 1, 1, 0], [1, 1, 1, 0, 1, 1, 0], [0, 1, 0, 1, 1, 1, 0], [1, 1, 0, 1, 1, 1, 0], [1, 0, 1, 1, 1, 1, 0], [0, 1, 1, 1, 1, 1, 0], [1, 1, 1, 1, 1, 1, 0], [1, 0, 1, 0, 1, 0, 1], [0, 1, 1, 0, 1, 0, 1], [1, 1, 1, 0, 1, 0, 1], [0, 1, 0, 1, 1, 0, 1], [1, 1, 0, 1, 1, 0, 1], [1, 0, 1, 1, 1, 0, 1], [0, 1, 1, 1, 1, 0, 1], [1, 1, 1, 1, 1, 0, 1], [0, 1, 0, 1, 0, 1, 1], [1, 1, 0, 1, 0, 1, 1], [1, 0, 1, 1, 0, 1, 1], [0, 1, 1, 1, 0, 1, 1], [1, 1, 1, 1, 0, 1, 1], [1, 0, 1, 0, 1, 1, 1], [0, 1, 1, 0, 1, 1, 1], [1, 1, 1, 0, 1, 1, 1], [0, 1, 0, 1, 1, 1, 1], [1, 1, 0, 1, 1, 1, 1], [1, 0, 1, 1, 1, 1, 1], [0, 1, 1, 1, 1, 1, 1], [1, 1, 1, 1, 1, 1, 1]] **(1)**

So here are the possible configurations.

```
> Configs:= [seq(seq([op(s),op(t)],t=Rowconfigs),s=Rowconfigs)]:
  nops(Configs);
```

1764 **(2)**

There are 1764 allowed configurations.
Consider the 1764×1764 transition matrix T such that $T_{ij} = 1$ if the top two rows of a 3×7 sub-array could be in configuration i while the bottom two rows are in configuration j (i.e. the middle row is compatible with both i and j), and 0 otherwise. The following Maple code computes it.

```
> compat:= proc(i,j) local k,col;
     if Configs[i][8..14] <> Configs[j][1..7] then return 0 fi;
     for k from 1 to 7 do
        col:= [Configs[i][k],Configs[i][k+7],Configs[j][k+7]];
        if col = [0,0,1] or col = [0,1,1] then return 0 fi;
     od;
     1
  end proc:
  T:= Matrix(1764,1764,compat);
```

(3)

$$T := \begin{bmatrix} \textit{1764 x 1764 Matrix} \\ \textit{Data Type: anything} \\ \textit{Storage: rectangular} \\ \textit{Order: Fortran\_order} \end{bmatrix} \quad (3)$$

Thus for $n \geq 2$, $a(n) = u^T T^{n-2} u$ where u is a 1764-dimensional column vector of all 1's.

```
> u:= Vector[column](1764, 1):
```

To check, here are the first few entries of our sequence for $n \geq 2$. We first recursively compute $T^n u$ for various n.

```
> Tnu[0]:= u:
  for n from 1 to 24 do Tnu[n]:= T . Tnu[n-1] od:
```

```
> seq(u^%T . Tnu[n], n=1..24);
```

10710, 65025, 221340, 753424, 1913940, 4862025, 10332630, 21958596, 41363322, 77915929, 134523480, 232257600, 375406920, 606784689, 931373730, 1429596100, 2104920510, 3099260241, 4409811252, 6274540944, 8675694300, 11995725625, 16187575950, 21844248804 **(4)**

Here is the empirical formula. It says that $u^T T^n Q(T)\, u = 0$ for all nonnegative integers n, where Q is the following polynomial.

```
> n:= 'n': empirical:=a(n) = 2*a(n-1) +6*a(n-2) -14*a(n-3) -14*a
  (n-4) +42*a(n-5) +14*a(n-6) -70*a(n-7) +70*a(n-9) -14*a(n-10)
  -42*a(n-11) +14*a(n-12) +14*a(n-13) -6*a(n-14) -2*a(n-15) +a
  (n-16);
  Q:= unapply(add(coeff((lhs-rhs)(empirical),a(n-i))*t^(16-i),i=0.
  .16),t);
```

$$empirical := a(n) = 2\, a(n-1) + 6\, a(n-2) - 14\, a(n-3) - 14\, a(n-4) + 42\, a(n-5) + 14\, a(n-6) - 70\, a(n-7) + 70\, a(n-9) - 14\, a(n-10) - 42\, a(n-11) + 14\, a(n-12) + 14\, a(n-13) - 6\, a(n-14) - 2\, a(n-15) + a(n-16)$$

$$Q := t \mapsto t^{16} - 2\, t^{15} - 6\, t^{14} + 14\, t^{13} + 14\, t^{12} - 42\, t^{11} - 14\, t^{10} + 70\, t^9 - 70\, t^7 + 14\, t^6 + 42\, t^5 - 14\, t^4 - 14\, t^3 + 6\, t^2 + 2\, t - 1 \quad (5)$$

It turns out that $Q(T)\, u = 0$. We check that this is the case.

```
> Qu:= add(coeff(Q(t),t,j)*Tnu[j],j=0..16):
> Qu^%T . Qu;
```

$$0 \quad (6)$$

This completes the proof.